
	INSTITUTO DOCTUM DE EDUCAÇÃO E TECNOLOGIA	
	CURSO: SISTEMA DE INFORMAÇÃO	
	PRÁTICA DE PROGRAMAÇÃO I	
	LISTA DE EXERCÍCIOS I	
	PROFESSOR: NATÁLIA DA MATA CAMPOS	
	ALUNO:	

APOSTILA - Portugol

Portugol: consiste em um conjunto de regras sintáticas e semânticas utilizadas para desenvolver uma solução algorítmica que, posteriormente, pode ser transcrita para uma linguagem de Programação. Para a prática de Algoritmos, utilizaremos as regras sintáticas e semânticas do Portugol WebStudio.

Estrutura Básica:

```

programa
{
    // Inclusões de bibliotecas, quando houver a necessidade de utiliza-las.
    funcao inicio()
    {
        /* Declarações de variáveis e constantes locais
        * Entradas
        * Processamento de dados.
        * Saídas.
        */
    }
}

```

Tipos de Dados:

- cadeia
- caracter
- inteiro
- real
- logico (falso ou verdadeiro)

Comandos:

- Atribuição: =
- Entrada: leia
- Saída: escreva
- Declaração de variáveis: Tipo de dado lista de variáveis
- Comentário:
 - // Comenta uma linha
 - /* Comenta um bloco: conjunto de linhas. */

Identificadores:

Para a formação de identificadores, faz-se necessário seguir as seguintes regras:

- O primeiro caracter deve ser sempre uma letra ou o sublinhado (_);
- Não são permitidos espaços em branco ou caracteres especiais (@, \$, +, %, etc.);
- Não podem ser usadas palavras reservadas.

Operações Aritméticas:

As operações aritméticas são utilizadas em Portugol com o mesmo objetivo da matemática: a realização de cálculos. Porém, com algumas modificações na simbologia. Veja o quadro abaixo:

Operação	Símbolo	Prioridade
Adição	+	1
Subtração	-	1
Multiplicação	*	2
Divisão	/	2
Resto da divisão inteira	%	2

Condicional:

```

se (condicao){
    //comandos
} senao se(condicao){
    //comandos
} senao{

```

```
//comandos
```

```
}
```

```
se (imc < 18.5){  
    escreva("Você está abaixo do peso.\n")  
} senao se (imc >= 18.5 e imc < 24.9){  
    escreva("Você está com peso normal.\n")  
} senao se (imc >= 24.9 e imc < 29.9){  
    escreva("Você está com sobrepeso.\n")  
}senao  
    escreva("Você está com obesidade.\n")
```

Estrutura de repetição LOOP

para (inicialização; condição; atualização)

Exemplos:

```
para(inteiro i = 1; i<=10; i++){  
  
    escreva("\n Bom dia – ", i)  
  
}
```

Exemplos:

```
programa {  
    funcao inicio() {  
        // declaração de variáveis  
        inteiro idade  
        real altura  
        cadeia nome  
        caracter inicial  
        logico exemplo  
        // atribuição de valores  
        inicial = 'V'  
        exemplo = verdadeiro  
        // entrada de dados  
        escreva ("Informe o nome:")  
        leia (nome)  
        escreva ("Informe a altura:")  
        leia (altura)  
        // saída de dados  
        escreva (nome)  
        escreva ("Altura", altura)  
    }  
}
```

DESCRIÇÃO: Faça um algoritmo em Portugol que leia 2 valores e calcule as 4 operações básicas sobre eles.

```
programa {  
    funcao inicio() {  
        real x, y, soma, sub, mult, div  
  
        escreva("Informe o primeiro número: ")  
        leia(x)  
        escreva("Informe o segundo número: ")  
        leia(y)  
  
        soma = x + y  
        sub = x - y  
        mult = x * y  
        div = x / y  
  
        escreva("\nSOMA: ", soma)  
        escreva("\nSUBTRAÇÃO: ", sub)  
        escreva("\nMULTIPLICAÇÃO: ", mult)  
        escreva("\nDIVISÃO: ", div, "\n")  
    }  
}
```

Exercícios

1. Faça um algoritmo em Portugol que leia o nome e salário de um funcionário, calcule e mostre o reajuste do salário com um aumento de 10%.

FICA A DICA!! O Desenvolvimento e organização do raciocínio lógico, bem como a compreensão da estrutura básica de algoritmos, estão associados a uma rotina de estudos e prática contínua.